

# Monte Carlo simulation of One Banana, Two Banana

#

# Copyright (c) 2010 John Graham-Cumming

#

# For game details see

# http://www.amazon.co.uk/Orchard-Toys-One-Banana-Game/dp/B001T3AAI4

#

# In the game there are a total of 54 cards of five types:

#

# 14 cards with three bananas on them

# 14 cards with two bananas on them

# 14 cards with one banana on them

# 6 cards with a swamp on them

# 6 cards with a banana skin

#

# The cards are placed face down on the table. At each turn the

# player can take as many cards as they wish. If they turn over a

# swamp or banana skin card they must stop; otherwise they stop when

# they choose.

#

# The total number of bananas is added up (call it X). If the banana

# skin card was turned over then the player moves BACK X spaces, if

# not the player moves FORWARD X spaces.

#

# Players have perfect knowledge of the state of the cards as they are

# placed face up once drawn. Once all six banana skin cards have be

# drawn the cards are all placed face down on the table again.



# The simulation runs through all the possible card positions and

# plays a large number of random draws for each possible number of

# cards a player might draw. It then keeps track of the score for a

# number of different card counting scenarios.

#

# Card counting scenarios:

#

# Number of skins: the player only keeps track of the number of banana

# skins that are on the table.

#

# Number banana cards: the player calculates the number of banana

# cards on the table.

#

# Number of three bananas: the player keeps track of the number of

# high scoring three banana cards on the table.



use strict;

use warnings;



# This is a two dimensional array. The first dimension is the number

# of skins (6, 5, 4, 3, 2, 1). The second dimension is the number of

# cards the player will pick. Each entry is a hash with keys score

# (the total score of all trials) and count (the total number of

# trials).



my @skins;



# Similar arrays to @skins, but here the first dimension is the number

# of cards on the table of the specific type.



my @bananas;

my @threes;



# Generate all possible valid board positions using nested loops.

# Note that the skin cards can never be zero because that's when the

# board resets.

#

# There are 6 x 7 x 15 x 15 x 15 possible board positions (115,248).



for my $sk (1..6) {

for my $sw (0..6) {

for my $on (0..14) {

for my $tw (0..14) {

for my $th (0..14) {



# Arbitrarily chosen run of 100 plays



for my $sim (0..99) {



# Allow the player to pick up to 10 cards



for my $to_pick (1..10) {



# The state of the board at any time can be represented by

# a 5-tuple (Skin, Swamp, One, Two, Three) giving the

# number of cards of each type present on the playing

# board. The initial state is (6, 6, 14, 14, 14).



my @cards = ( $sk, $sw, $on, $tw, $th );



my $score = 0;

for my $pick (1..$to_pick) {

my $total = $cards[0] + $cards[1] + $cards[2] + $cards[3]

+ $cards[4];

my $picked = int(rand($total));

if ( $picked < $cards[0] ) {

$score = -$score;

last;

} elsif ( $picked < ( $cards[0] + $cards[1] ) ) {

last;

} elsif ( $picked < ( $cards[0] + $cards[1] + $cards[2] ) ) {

$cards[2]--;

$score += 1;

} elsif ( $picked < ( $cards[0] + $cards[1] + $cards[2] +

$cards[3] ) ) {

$cards[3]--;

$score += 2;

} else {

$cards[4]--;

$score += 3;

}

}



$skins[$sk][$to_pick]{score} += $score;

$skins[$sk][$to_pick]{count}++;



$bananas[$on+$tw+$th][$to_pick]{score} += $score;

$bananas[$on+$tw+$th][$to_pick]{count}++;



$threes[$th][$to_pick]{score} += $score;

$threes[$th][$to_pick]{count}++;

}

}

}

}

}

}

print "$sk

";

}





if ( open F, ">skins.csv" ) {

for my $i ( 1..6 ) {

for my $k (1..10) {

print F "$i,$k,$skins[$i][$k]{score},$skins[$i][$k]{count}

";

}

}

close F;

}







Banana Skins Optimum Number of Cards Expected Score

1 8 5.1

2 6 3.4

3 4 2.6

4 3 2.2

5 2 1.9

6 2 1.7





And also, if you are player and have exceeded the expected score with fewer than the recommended cards you'll know it's best to stop.

The children's game One Banana, Two Banana is a high stakes game of probability theory in action. Or something like that. Actually, it's a fun game where you have to take probability into account when deciding what to do.Conceptually, the game is simple. There are 54 cards of five types. Three of the card types have bananas on them (one, two or three bananas), one card type has a banana skin on it and one card type has a swamp on it. All the cards are placed face down on the table and shuffled about. Each player takes turns at turning over as many cards as they want.The total distance they move on the board (winning is a simple first past the post system) is the sum of the number of bananas on the banana cards. But if they pick a banana skin card then they stop picking cards and must reverse direction by the sum of the number of bananas on the banana cards. The swamp card simply stops a turn without a penalty.There are six banana skin cards to start with and as they are removed from the pack they are placed on a special card for all to see. At every stage everyone knows the number of banana skin cards remaining. Thus you can adjust your strategy based on the number of banana skins that have caused others to go back.So I wrote a little program that simulates One Banana, Two Banana games (or at least board positions) and see what the expected score is depending on the number of cards that the player chooses to pick. I assume a very simple strategy of deciding up front how many cards to pick. A more complexity strategy would be to look at the score as the player turns over cards and decide when to quit.First, here's the code:And here's a chart showing the average score from the Monte Carlo simulation for each of the six possible numbers of banana skins on the board. The X axis shows the number of cards picked (or desired to be picked), and the Y axis the average score.If you look just at the maximum scores for each of the banana skin counts, then a simple pattern emerges.So, just memorize that pattern and you'll be a better player.I've removed that sentence because it is not correct. That's a different strategy based on stopping based on the cards you have already turned over; that deserves a separate analysis to find the best strategy.Watch out Vegas, here I come.